

Building on UG ex 5.13, the following latent variable example gives a moderation plot that describes the interaction between f1 and f2 in their influence on f3. The regression equation is

$$f3 = b1*f1 + b2*f2 + b3*f1*f2 + e. \quad (1)$$

Let's say that we view the f1, f2 interaction such that f2 moderates the influence of f1 on f3 (one can also express this as f1 moderating the effect of f2 on f3). This means that we rewrite the regression in (1) as

$$f3 = (b1 + b3*f2)*f1 + b2*f2 + e. \quad (2)$$

The effect of f1 on f3 as moderated by f2 can be plotted by showing the influence of f1 on f3 for different f2 values.

The Mplus input is as follows. Exclamation marks have been added to indicate lines that have been added or modified relative to ex 5.13.

```
TITLE:      this is an example of a SEM with
            continuous factor indicators and an
            interaction between two latent variables
```

```
DATA: FILE IS ex5.13.dat;
```

```
VARIABLE:  NAMES ARE y1-y12;
```

```
ANALYSIS:  TYPE = RANDOM;
```

```
           ALGORITHM = INTEGRATION;
```

```
MODEL:     f1 BY y1-y3* ; !
```

```
           f2 BY y4-y6* ; !
```

```
           f1-f2@1 ; !
```

```
           f3 BY y7-y9 ;
```

```
           f4 BY y10-y12 ;
```

```
           f4 ON f3 ;
```

```
           f3 ON f1 (b1) !
```

```
           f2 (b2) ; !
```

```

f1xf2 | f1 XWITH f2;

f3 ON f1xf2 (b3); !

OUTPUT:   TECH1 TECH8;

Model Constraint: !

PLOT (lowf2 highf2); !

LOOP(f1, -3, 3, 0.1); !

lowf2 = (b1+b3*(-1))*f1+b2*(-1); !

highf2 = (b1+b3*(+1))*f1+b2*(+1); !

Plot: !

type = plot2; !

```

The lowf2 and highf2 expressions correspond to (2) for f2 values of -1 and +1. This gives a plot of the effect of f1 on f3 at 1 SD below and 1 SD above the zero mean of f2, plotted over the f1 range of -3 SD to +3 SD. This is similar to Figure 3 of the FAQ Latent variable interactions. Note that the metrics of f1 and f2 have been set by fixing their variances at 1. This makes the f2 values of -1 and +1 correspond to 1 SD below and 1 SD above the zero mean of f2.

The plot is obtained from the LOOP plots option of the Plot menu. 95% confidence bands are also given.

